// 编译：240884432
VIP15策略说明：
本策略是基于SDO（Swing Direction Oscillator）指标的交易系统。
主要功能包括：
1. SDO指标计算
2. 入场信号生成
3. 交易执行（多头和空头）
4. 跟踪止损
策略的核心思想是利用SDO指标识别市场超买超卖状态，并在价格突破时入场，
同时使用跟踪止损来管理风险。
*/

// 参数定义
Params
    Numeric Period(20);        // 周期
    Numeric Pds(10);           // 平滑周期
    Numeric TrailingStopRate(30); // 跟踪止损比例 (5-100)
    Numeric Fund(20000);       // 投入保证金

		//Bar更新事件函数，参数indexs表示变化的数据源图层ID248844320数组
/*
变量说明：
这里定义了策略所需的各种变量，包括：
- SDO相关变量：用于计算和存储SDO值
- 交易信号相关变量：用于生成和存储交易信号
- 持仓管理相关变量：用于跟踪止损和仓位管理
*/
// 变量定义
Vars
    Numeric OverBought(40);    // 超买水平
    Numeric OverSold(-40);     // 超卖水平
    Numeric LBPeriod(50);      // 回溯周期
    Numeric DVal(0);           // 差值
    Numeric DDVal(0);          // 最终差值
    Numeric Dist(0);           // 距离
    Numeric SDOVal(0);         // SDO值
    Series<Bool> cond1;        // 条件1：SDO穿越超买线
    Series<Bool> cond2;        // 条件2：SDO穿越超卖线
    Series<Numeric> HH;        // 最高价
    Series<Numeric> LL;        // 最低价
    Series<Numeric> DD;        // 多头入场价
    Series<Numeric> SS;        // 空头入场价
    Series<Numeric> KG;        // 开关信号
    Series<Numeric> Barcout;   // 计数器
    Numeric Lots(0);           // 交易手数

    // 跟随出场相关变量
    Series<Numeric> HighestLowAfterEntry;  // 开仓后出现的最高的最低价
    Series<Numeric> LowestHighAfterEntry;  // 开仓后出现的最低的最高价    
    Series<Numeric> liQKA(1);              // 流动性系数
    Series<Numeric> DliqPoint;             // 多头止损点
    Series<Numeric> KliqPoint;             // 空头止损点
    Series<Numeric> entBar;                // 入场Bar
    Series<Numeric> entPrice;              // 入场价格
    
    Plot plt1;                 // 绘图对象

/*
函数定义：
这里定义了策略使用的主要函数，包括：
1. Trailing_Stop: 计算跟踪止损
2. Initialize_Vars_On_Entry: 开仓时初始化变量
3. Reset_Vars_On_Exit: 平仓时重置变量
这些函数用于管理交易过程中的各种状态和计算。
*/
// 函数定义
Defs
    // 跟随出场：计算入场后的最高的最低价、最低的最高价, 及对应的止损线
    Integer Trailing_Stop(Numeric Enable_Display = 1)
    {
        // 如果是多头持仓
        If(MarketPosition > 0)
        {
            If(CurrentBar == entBar)
            {
                HighestLowAfterEntry = L;  // 初始化为当前最低价
            }
            Else
            {
                HighestLowAfterEntry = MAX(HighestLowAfterEntry, L);  // 更新最高的最低价
            }
        }
        // 如果是空头持仓
        Else If(MarketPosition < 0)
        {    
            If(CurrentBar == entBar)
            {
                LowestHighAfterEntry = H;  // 初始化为当前最高价
            }
            Else
            {
                LowestHighAfterEntry = MIN(LowestHighAfterEntry, H);  // 更新最低的最高价
            }
        }
        
        Commentary("LowestHighAfterEntry"+text(LowestHighAfterEntry));
        Commentary("HighestLowAfterEntry"+text(HighestLowAfterEntry));
        
        // 计算止损点
        DliqPoint = HighestLowAfterEntry - (Open * TrailingStopRate /10000) * liQKA;
        KliqPoint = LowestHighAfterEntry + (Open * TrailingStopRate /10000) * liQKA;
        
        // 如果启用显示，可以在这里添加绘图代码
        If(Enable_Display == 1)
        {
            // 绘图代码已被注释掉
            //If(MarketPosition > 0)
                //PlotAuto("DliqPoint", DliqPoint, DliqPoint, Green, Enum_Cross);
            //If(MarketPosition < 0)
                //PlotAuto("KliqPoint", KliqPoint, KliqPoint, Magenta, Enum_Cross);
        }
        
        Return 1;
    }

    // 开仓时初始化变量
    Numeric Initialize_Vars_On_Entry()
    {
        entBar = CurrentBar;
        entPrice = EntryPrice;
        liQKA = 1;
        
        Return 1;
    }

    // 平仓时重置变量
    Numeric Reset_Vars_On_Exit()
    {
        entBar = 0;
        entPrice = 0;
        
        Return 1;
    }

/*
主要事件处理：
OnBar函数是策略的核心，在每个新的K线到来时执行。
主要步骤包括：
1. 计算SDO值
2. 生成交易信号
3. 执行交易（开仓和平仓）
4. 管理持仓（包括跟踪止损）
*/
// 主要事件处理
Events
    OnBar(ArrayRef<Integer> indexs)
    {
        // 计算交易手数
        Lots = Max(1, Round(Fund/(O*ContractUnit*BigPointValue* MarginRatio/rollover), 0));
        
        // 计算SDO值
        Dist = Abs(C - C[Period]);
        DVal = (Dist - Lowest(Dist, Period)) / (Highest(Dist, LBPeriod) - Lowest(Dist, LBPeriod));
        If (C > C[Period])
        {
            DDVal = DVal;
        }
        Else If (C < C[Period])
        {
            DDVal = -DVal;
        }
        Else
        {
            DDVal = 0;
        }
        SDOVal = Average(DDVal, Pds) * 100;

        // 绘制SDO、超买和超卖水平
        PlotNumeric("SDO", SDOVal);
        PlotNumeric("OverBought", OverBought);
        PlotNumeric("OverSold", OverSold);

        // 定义交易条件
        cond1 = CrossUnder(SDOVal, OverBought);  // SDO下穿超买线
        cond2 = CrossOver(SDOVal, OverSold);     // SDO上穿超卖线
        HH = Highest(H, Period);
        LL = Lowest(L, Period);
        
        // 如果满足交易条件，重置计数器和相关变量
        if(cond1 or cond2)
        {
            DD = HH;
            SS = LL;
            KG = 1;
            Barcout = 0;
        }
        if(KG > 0)
        {
            Barcout = Barcout + 1;
        }
        
        // 多头入场条件
        if(MarketPosition != 1 and KG[1] > 0 and H >= DD[1] and DD[1] > 0)
        {
            Buy(Lots, max(open, DD[1]));
            KG = 0;
            Barcout = 0;
            HighestLowAfterEntry = max(open, DD[1]); 
            Initialize_Vars_On_Entry();
        }

        // 空头入场条件
        if(MarketPosition != -1 and KG[1] > 0 and L <= SS[1] and SS[1] > 0)
        {
            SellShort(Lots, min(open, SS[1]));
            KG = 0;
            Barcout = 0;
            LowestHighAfterEntry = min(open, SS[1]);
            Initialize_Vars_On_Entry();
        }    
        
        // 如果计数器超过平滑周期，重置相关变量
        if(Barcout > Pds)
        {
            Barcout = 0;
            DD = 0;
            SS = 0;
            KG = 0;
        }    
        
        Commentary("Barcout"+text(Barcout));
        Commentary("DD"+text(DD));
        Commentary("SS"+text(SS));
        
        // 计算跟踪止损线
        Trailing_Stop(1);     // 参数 == 1 时，在主图绘制止损线
        
        // 多头跟踪止损
        If(MarketPosition > 0 And L <= DliqPoint[1] And CurrentBar > entBar)
        {
            Sell(0, Min(Open, DliqPoint[1]));
            KG = 0;
            Commentary("多头跟踪止损");
            Reset_Vars_On_Exit();
        }
        
        // 空头跟踪止损
        If(MarketPosition < 0 And H >= KliqPoint[1] And CurrentBar > entBar)
        {
            BuyToCover(0, Max(Open, KliqPoint[1]));
            KG = 0;

            Reset_Vars_On_Exit();
        }        
    }

